



PRODUCT NAME

TM58XX

TITLE

TM58XX comparator test temperature

APPLICATION NOTE

1. explanation
2. program
3. application circuit diagram

Use IC: TM58P10
 Crystal oscillator: 4MHZ
 WATCHDOG: DISABLE
 Mode: Advanced

Function : utilize the Charging and discharging electric to realize the function of test temperature.

The range of test temperature: 0~40'C, test temperature is displayed by two digital tube. Measure precision: +/-1'C. The sensor of using: the temperature sensor of AT-103 type. In the program ,it tests temperature 3 times percent second, refurbishes the temperature displaying 1 time percent second. In the program, it sets sensor self-check function. When sensor cuts or short-circuits, digital tube second flash displays 'E', it shows error. After treats replaces the sensor and on the electricity, the system may restore normally again. In the situation of the system starting machine 4s intrinsic sensor has not made a mistake, the digital tube displays '88', but doesn't display measures the test temperature result, it shows that the digital tube displays normally. In the program ,the processing way to test temperature result is as follows:

The count value of carries on the RC charge and discharge electric to the sensor (16 digits, storage in (buff3, buff4) contrast to the value of carries on the RC charge and discharge electric to the resistance (16 digits, storage in (buff1, buff2), carries on 16 subtraction (namely (buff3, buff4) - (buff1, buff2)). The result of the subtraction storage in (buff3, buff4). We can get the measure temperature value through the result of the measure count value of RC charge and discharge electric cancellation.

```

;-----
;Title:TM58xx Application note
;file name----comparator MEASURE_T.asm
;
;Revision History:
;Rev:    Date:    Reason:
;1.0    5/11/2004    original
;
;-----
;
;-----
indf      equ    00h
tmr0     equ    01h
pc       equ    02h
status   equ    03h
fsr      equ    04h
porta    equ    05h
portb    equ    06h
wake_up  equ    20h
irqm     equ    21h
irqf     equ    22h
;-----
;porta      I/O define
;-----

```

```

led_com2    equ    0
led_com1    equ    1
s_in        equ    2
s_out       equ    3
;-----
;irqm bits define
intm        equ    7
exintm      equ    2
tmr0m       equ    0
;irqf bits define
tmr0f       equ    0
;status bits define
c           equ    0
z           equ    2

w           equ    0
f           equ    1
;-----
;ctr_flag bits define
display     equ    0    ; Digital tube high-low digit display control symbol,
                       ;"0":display the high digit,"1":display the low digit
oc_crt      equ    1    ; measure temperature control symbol
rt_ct_crt   equ    2    ; Test sensor and compareing resistance controlsymbol,
                       ;"0":Sensor,"1": Compareing resistance
error_crt   equ    3    ; the sensor error control symbol
error_show  equ    4    ; error imformation'E'display control symbol
data_ctr    equ    5    ; test temperature data processing control symbol
add_time    equ    6    ; Needs to carry on the capacitance charge electronic to count
                       ;the symbol
add_out     equ    7    ;the symbol of capacitance charges the sufficiently
;-----
done        equ    0
;-----
; position definition
;-----
ctr_flag    equ    08h  ;display the symbol of measure temperature control
a_temp      equ    09h  ; The protection temporarily stores of accumulator A
status_temp equ    0ah  ; The protection temporarily stores of condition register Status
ms4_acc     equ    0bh  ; Accumulator,uses in the 4ms time
sec_acc     equ    0ch  ; second counter
time_cnt    equ    0dh  ; Opening machine 4s did not display the time counter of
                       ;the measures temperature data

dis_h       equ    0eh  ; The temperature high digit displays the section code
                       ;temporarily stores
dis_l       equ    0fh  ; The temperature low digit displays the section code
                       ;temporarily stores

porta_buf   equ    10h  ;porta buffer

```

```

oc_value    equ    11h    ; the measure temperature value temporarily stores
buffer1     equ    12h    ; the low 8 digit of RC charges and discharges count value
buffer0     equ    13h    ; the high 8 digit of RC charges and discharges count value
buff1      equ    14h    ;buff1...buff4:RC 充放电 count 值与表格数据比较暂存
buff2      equ    15h
buff3      equ    16h
buff4      equ    17h
flag       equ    18h

byte       equ    1ah
r0        equ    1bh
r1        equ    1ch
temp      equ    1dh
count     equ    19h
;
;-----
;prog. start here
;-----
;-----
    org    3ffh
    lgoto  main
    org    3feh
    lgoto  time_int
    org    000h
;=====
temperature_h:
    addam  pc,1
    retla  0    ;"0'c"----R:27.28k
    retla  0    ;"1'c"----R:26.13K
    retla  0    ;"2'c"----R:25.03K
    retla  0    ;"3'c"----R:24.00K
    retla  0    ;"4'c"----R:23.00K
    retla  0    ;"5'c"----R:22.05K
    retla  0    ;"6'c"----R:21.15K
    retla  0    ;"7'c"----R:20.30K
    retla  0    ;"8'c"----R:19.48K
    retla  0    ;"9'c"----R:18.70K
    retla  0    ;"10'c"----R:17.96K
    retla  0    ;"11'c"----R:17.24K
    retla  0    ;"12'c"----R:16.65K
    retla  0    ;"13'c"----R:15.90K
    retla  0    ;"14'c"----R:15.28K
    retla  0    ;"15'c"----R:14.69K
    retla  0    ;"16'c"----R:14.12K
    retla  0    ;"17'c"----R:13.58K
    retla  0    ;"18'c"----R:13.06K
    retla  0    ;"19'c"----R:12.56K
    retla  0    ;"20'c"----R:12.09K
    retla  0    ;"21'c"----R:11.63K

```

```

retla 0 ;"22'c"----R:11.20K
retla 0 ;"23'c"----R:10.78K
retla 0 ;"24'c"----R:10.38K
retla 0 ;"25'c"----R:10.00K
retla 0 ;"26'c"----R:9.632K
retla 0 ;"27'c"----R:9.281K
retla 0 ;"28'c"----R:8.944K
retla 0 ;"29'c"----R:8.622K
retla 0 ;"30'c"----R:8.313K
retla 0 ;"31'c"----R:8.014K
retla 0 ;"32'c"----R:7.728K
retla 0 ;"33'c"----R:7.454K
retla 0 ;"34'c"----R:7.192K
retla 0 ;"35'c"----R:6.940K
retla 0 ;"36'c"----R:6.699K
retla 0 ;"37'c"----R:6.467K
retla 0 ;"38'c"----R:6.245K
retla 0 ;"39'c"----R:6.032K
retla 0 ;"40'c"----R:5.827K

```

```

=====
temperature_1:

```

```

addam pc,1
retla 46h ;"0'c"
retla 48h ;"1'c"
retla 4bh ;"2'c"
retla 4dh ;"3'c"
retla 50h ;"4'c"
retla 53h ;"5'c"
retla 56h ;"6'c"
retla 59h ;"7'c"
retla 5ch ;"8'c"
retla 5eh ;"9'c"
retla 63h ;"10'c"
retla 66h ;"11'c"
retla 68h ;"12'c"
retla 6bh ;"13'c"
retla 6fh ;"14'c"
retla 73h ;"15'c"
retla 76h ;"16'c"
retla 79h ;"17'c"
retla 7dh ;"18'c"
retla 81h ;"19'c"
retla 84h ;"20'c"
retla 87h ;"21'c"
retla 8bh ;"22'c"
retla 8eh ;"23'c"
retla 92h ;"24'c"
retla 96h ;"25'c"

```

```

retla 9bh ;"26'c"
retla 0a0h ;"27'c"
retla 0a5h ;"28'c"
retla 0a8h ;"29'c"
retla 0abh ;"30'c"
retla 0b0h ;"31'c"
retla 0b5h ;"32'c"
retla 0b8h ;"33'c"
retla 0bch ;"34'c"
retla 0c0h ;"35'c"
retla 0c6h ;"36'c"
retla 0cah ;"37'c"
retla 0d0h ;"38'c"
retla 0d4h ;"39'c"
retla 0e0h ;"40'c"

```

```

=====
;DIGI_LED display section code table
=====

```

```
dis_table:
```

```

addam pc
retla 40h ;"0"
retla 79h ;"1"
retla 24h ;"2"
retla 30h ;"3"
retla 19h ;"4"
retla 12h ;"5"
retla 02h ;"6"
retla 78h ;"7"
retla 00h ;"8"
retla 10h ;"9"

```

```

=====
; 8 digit Binary count change to BCD code subprogram
=====

```

```
btobcd8bit:
```

```

bcm status,c
movla 08h ; around shift 8 times
movam count
clrm r0
clrm r1

```

```
loop8:
```

```

rlm byte
rlm r1
rlm r0 ; Byte shifts to the left into r1, r0
decmsz count
lgoto adjdec ; carries on the result of r0, r1 shifting digit to the
; bcd adjustment

```

```
ret
```

```
adjdec:
```

```
movla r1
```

```

    movam  fsr
    lcall  adjbcd
    movla  r0
    movam  fsr
    lcall  adjbcd
    lgoto  loop8
;-----
adjbcd:
    movla  03h
    addam  indf,w           ;Is low half byte+3>7?
    movam  temp
    btmsc  temp,3
    movam  indf           ;yes,low half byte +3>7,lsd+3--->lsd
    movla  30h
    addam  indf,w           ; Is high half byte+3>7?
    movam  temp
    btmsc  temp,7
    movam  indf           ;yes, high half byte+3>7,msd+3--->msd
    ret
;=====
;main program
;=====
main:
    clrm   time_cnt
    clrm   sec_acc
    clrm   ctr_flag
    clrm   ms4_acc
    bsm    ctr_flag,oc_crt
    clrm   dis_l           ; when boot-strap,the digital tube display'88'
    clrm   dis_h
    movla  0ffh
    movam  portb           ;portb export '0ffh'
    clra
    ;clear Accumulator
    iodir  portb           ; set portb as output port
    movla  03h
    movam  porta           ;portae export '03h'
    movam  porta_buf
    movla  b'00000100'
    iodir  porta           ; set porta2 as input,the other as output
    clrm   tmr0
    movla  00h
    select
    bcm    irqf,   tmr0f
    bsm    irqm,   intm
    bsm    irqm,   tmr0m
;-----
read_time:
    movla  64h           ; judge whether 100*40us is equal to 4ms
    xoram  ms4_acc,w

```

```

    btms  status,z
    lgoto read_time ;If 4ms is not arrive,then continue to judge time
;-----
display_oc: ; If 4ms is arrive,then display temperature program section
    clrm  ms4_acc ;clear ms4_acc
    btms  ctr_flag,display; judge digital tube should display high digit or low digit?
    lgoto show_h
; -----
show_l: ; digital tube display low digit
    bcm  ctr_flag,display ; clear digital tube high and low digit display
        ;control symbol,let it point to high digit

    bsm  porta_buf,led_com2 ; turn off high digit digital tube
    movm porta_buf,w
    movam porta
    movm dis_l,w
    movam portb ; send the low digit display section code to portb
    bcm  porta_buf,led_com1 ; turn on the low digit digital tube
    movm porta_buf,w
    movam porta
    lgoto deal_time ; change to the timing time-processing program section
; -----
show_h: ; digital tube display the high digit
    bsm  ctr_flag,display ;clear the control symbol of the high and low digit of
        ;digital tube ,let it points to the low digit

    bsm  porta_buf,led_com1 ; turn off the low digit digital tube
    movm porta_buf,w
    movam porta
    movm dis_h,w
    movam portb ; send the high digit displaying section code to portb
    bcm  porta_buf,led_com2 ; turn on the high digit digital tube
    movm porta_buf,w
    movam porta
;-----
deal_time: ;the program section of timing time-processing
    incm  sec_acc
    btms  ctr_flag,error_crt ; judge whether the sensor is error
    lgoto sensor_no_error ; If the sensor does`t error,then change to 1s time
        ;program section
; -----
sensor_error: ; If the sensor has something wrong,then go to the 0.5s
        ;time processing
        ;Does 125*4ms=500ms arrive?

    movla 7dh
    Xoram sec_acc,w
    btms  status,z
    lgoto read_time ; If 0.5s does`t arrive,then continue to judge time.
    clrm  sec_acc ; If 0.5s arrive,clear sec_acc
    btms  ctr_flag,error_show ; judge whether display error information 'E'
    lgoto set_e ;'E' is not at the display state,change to set 'E' to
        ;display program section

```

```

; -----
clear_e:                                ; If 'E' is at the display state,set 'E' as not display.
    bcm    ctr_flag,error_show          ; clear error information'E' display control symbol ,make
                                        ;it point to not display state

    movla  0ffh
    movam  dis_1                        ; send '0ffh' section code to dis_1,makes 'E' not display
    lgoto  read_time                    ; continue to judge time
; -----
set_e:                                  ;'E' is`t at the display state,set 'E' to display
    bsm    ctr_flag,error_show          ; set error information 'E' display control symbol to poin
                                        ; to display state

    movla  06h
    movam  dis_1                        ; send '06h' section code to dis_1,makes 'E'display
    lgoto  read_time                    ; continue to judge time
; -----
sensor_no_error:                        ; sensor is not error,go to 1s time processing
    movla  0fah                          ;Does 250*4ms=1000ms arrive?
    xoram  sec_acc,w
    btms   status,z
    lgoto  $+.12
    bsm    ctr_flag,oc_crt               ;set to measure temperature control symbol,make the
                                        ;measure temperature efficiency

    clrm   sec_acc                       ; If 1s arrives,clear sec_acc
    btmsc  ctr_flag,data_ctr            ; judge measure temperature data processing control
                                        symbol is inefficacy

    lgoto  $+8                           ;efficiency
    incm   time_cnt                     ; inefficacy
    movla  04h                           ; judge the time of 4*1s=4s measure teperature data not
                                        ; processing whether has arrived

    xoram  time_cnt,w
    btms   status,z
    lgoto  $+3                           ;4s not arrive
    clrm   time_cnt                     ;4s arrive
    bsm    ctr_flag,data_ctr
;-----
    btms   ctr_flag,data_ctr
    lgoto  read_time
test_tempratuer:                        ; the program section of measure temperature
    btms   ctr_flag,oc_crt              ; judge measure temperature whether is efficiency
    lgoto  read_time                    ; measure temperature is not efficiency,then continue
                                        ;to judge time

test_rt:                                 ; measure temperature is efficiency,start to test
                                        ;temperature

    btmsc  ctr_flag,error_crt
    lgoto  read_time

    btmsc  ctr_flag,add_time            ; Whether it is time?

```

```

lgoto    $+.10
btmsc   ctr_flag,add_out
lgoto    $+8
clrm    buffer0
clrm    buffer1
bsm     porta_buf,s_out    ; capacitance is charging
movm    porta_buf,w
movam   porta
bsm     ctr_flag,add_time  ; set time symbol
lgoto    read_time

btms    ctr_flag,add_out   ; Whether has finished count?
lgoto    read_time
bcm     ctr_flag,add_out   ; If count finished,clear the count symbol
bcm     ctr_flag,add_time
bcm     porta_buf,s_out    ; capacitance discharge
movm    porta_buf,w
movam   porta
;-----
; the count value of test by IC displaying
;-----
movm    buffer0,w          ; judge whether has error,if not renovates the temporary
                                ; storage device

btms    status,z
lgoto    no_err
movla   06h
subam   buffer1,w
btmsc   status,c          ;buffer1<=6, is measure temperature circuit short-circuit
lgoto    no_err
;have_error:                ; measure temperature circui error self-diagnose course
bsm     ctr_flag,error_crt ; sensor is cut,set sensor error control symbol
movla   0ffh
movam   dis_h
movla   06h
movam   dis_l
clrm    sec_acc
bsm     ctr_flag,error_show
lgoto    read_time
no_err
movm    buffer0,w          ;Renovates the temporary storage device
movam   buff3
movm    buffer1,w
movam   buff4

btms    ctr_flag,data_ctr  ; judge whether measure temperature data
                                ;processing control symbol is efficiency
lgoto    read_time          ;symbol is not efficiency,continue to judge time

```

;

```

get_oc:                                     ; By measured the temperature data which looks up
                                           ; the table to result the segment
        clrm    oc_value
loopz1:
        movm   oc_value,w
        lcall  temprature_h
        movam  buff3          ;buff1
        movm   oc_value,w
        lcall  temprature_l
        movam  buff4          ;buff2
        movm   buffer0,w
        movam  buff1          ;buff3
        movm   buffer1,w
        movam  buff2          ;buff4
        lcall  sub
        btmsc  status,c
        lgoto  get_oc_code
loopz2:
        incm   oc_value
        movla  28h            ;40'C
        subam  oc_value,w
        btmsc  status,c
        lgoto  loopz1        ;oc_value<40'C
        movla  28h
        movam  oc_value      ;oc_value>=40'C,then give 28h to oc_value
; -----
get_oc_code:                               ; the displaying segment of measure temperature
                                           ;process to program section
        movm   oc_value,w
        movam  byte
        lcall  btobcd8bit
        movm   r1,w
        andla  0fh
        lcall  dis_table
        movam  dis_l
        movm   r1,w
        andla  0f0h
        btmsc  status,z
        lgoto  get_hbit_code
        movla  0ffh
        movam  dis_h
        lgoto  read_time
get_hbit_code:
        movam  byte
        swapm  byte,w
        lcall  dis_table
        movam  dis_h
        lgoto  read_time
;

```

```

;
;=====
;      org      200h
;=====
; 16 digit no symbol Binary count subtraction subprogram
; The program realizes rtalh (high 8),Buff4 (low 8) expressed 16 binary numbers
;and the 16 binary numbers which expressed by Buff1 (high 8),Buff2 (low 8)
;cancellations,the result of reduces puts in rtalh (high 8) and rtall (low 8) center.
;(buff3,buff4)=(buff3,buff4)-(buff1,buff2)
; Because with reducing the no symbols binary number, the binary number which is
;denotated by (buff3,buff4) should be more than that is denotated by (buff1,buff2) ,
;or the outcome will be error.
;=====
sub:
; the low 8 digit cancellation
    movm  buff2,w
    subam buff4,f
    btms  status,c           ; judge whether rtall is able to reduce rhall?
    lgoto have_carry        ;rtall 不够减,向 rhall 有借位,转去执行 rtalh 减 1
decrease::高 8 位相减
    movm  buff1,w
    subam buff3,f
    ret
have_carry::高 8 位减 1
    movla 01h
    subam buff3,f
    btmsc status,c           ; judge rtalh reduce 1 is able?
    lgoto decrease          ; is able
    ret                       ; is not able,the result error,can`t go on reducing
;=====
;tmr0 interrupt subprogram(interrupt 1 time per 40us)
;=====
time_int:
    movam a_temp             ; Sends in a_temp the accumulator content to carry on
                             ;the protection
    swapm status,w           ; Exchanges the result the condition register status height
                             ; half byte content to deliver in mutually the accumulator
    movam status_temp        ; finally send the result into the status_temp to carry on
                             ;the ;protection
    movla 0E6h               ;set tmr0 value is "0E6h"(216+7+7=230),make
                             ;tmr0 interrupt 1 time per 100uS(256-156=100)
                             ;Which in order to compensate lgoto time_int~movam the
                             ;tmr0 6 instructions (consumes time is 7 instruction
                             ;period) and the hardware response interrupt consumes
                             ;the time (consumes time 7 instruction period),For tmr0
                             ;rewiring starting value when added 14 count values.

    movam tmr0
    incm  ms4_acc             ;4ms time

```

```

    btmsz   ctr_flag,add_time   ; whether need count charging time?
    lgoto   end_int             ;'NO'
    btmsz   porta,s_in         ; judge whether capacitance is full?
    lgoto   $+6
    incmsz  buffer1            ; electricity is not full,count value count
    lgoto   end_int
    incmsz  buffer0
    lgoto   end_int

    nop
    bcm     ctr_flag,add_time   ;capitance has abounded or overflow
    bsm     ctr_flag,add_out

end_int:
    bcm     irqf,tmr0f         ; clear tmr0 interrupt request flag
    swapm   status_temp,w     ; Exchanges the result the status_temp register high half
                                ;byte content to deliver in mutually the accumulator
    movam   status            ; send the accumulator content to condition register
                                ; status, restores the status condition before interrupt
    swapm   a_temp,f          ; Exchanges mutually the a_temp high half byte
                                ; content,The result sends into a_temp
    swapm   a_temp,w          ; Exchanges mutually the a_temp high half byte
                                ; content,The result sends in the accumulator, restores
                                ; the accumulator content before interruption
    reti     ; The interruption will return, after carries out this
                                ; instruction, the program will automatically open the
                                ; interruption
;
;
;=====
    end
;=====
;example
;=====

```

;During the time after the system electrify 4s,the digital tube displays '88',after 4s,displays the value of measure temperature,if the sensor`s resistance value is 11.20K(the resistance value of 22 ' C corresponding),then the temperature value should be '22'.In th any time when the program running,if the sensor is cut or short-circuit or 或比对电阻,then the digital tube second flash display 'E',express the sensor circuitry make some mistake.

